
Part 3 of  Building building “stuff” in POVRay 
OK, our tower is rather plain, just a pigment – Grey. It's 
time to address that.
In order to do that we need to think a little about our 
tower. What we want to do, for consistency, is to “de 
-pigment” all the stuff we want to “dress up”. Meaning, we 
need to remove the local pigment statements.

So, leaving the “woody stuff” aside for now, let's just 
make all the Grey stuff look more like masonry. SO, here 
is a version of the code where I've removed all the Grey 
pigment statements, except the one used in the tower call 
command.

Side note: I have been using the pigment “Grey” as it appears to be one of the few 
concessions POVRay makes to the English spelling of that colour. You can use “Gray” if you 
like! You'll also find that other “greyscale” colours are available and they use the prefix “Gray” 
too! “Gray10” is a darker (~10% white) grey, “Gray90” is a lighter shade etc.

#include "colors.inc"
#include "gridsetc.inc"

// switches and stuff
#declare axes = 0.1;
#declare noshadow = 1;
#declare grid = 0.05;
#declare rd=seed(31574);
#declare var = 1;
#declare Space = < 0, 1000, 0>;

// camera location
camera {
  location  < 25.0, 50, -50.0>
  look_at   <0.0, 6.0,  0.0>
  angle      20
}

// illumination
light_source { < 200,400,-100> White}
light_source { <-200, 50,-400> White}

background {SkyBlue}

plane { y, 0.0 pigment{ForestGreen} }

#if (axes > 0) object{axes1(axes,noshadow) translate < 0, 0.0, 0> } #end
#if (grid > 0) object{gridsp(grid) translate < 0, 0.0, 0> } #end  

// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ########## 

// ARROW SLIT CUTOUT
#declare wedgie = difference{
    box { <-1, -2, -4>, <1, 4, 0> }//pigment{Red} }
    
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y }// pigment{Blue} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y }// pigment{Blue} }

    box { <-2,  -3, -5>, <2, 0, 1> rotate -25*x}// pigment{Blue} }
}
    
// arched door, 1m wide, 2m high. Translate +1*y to put base at floor level
#declare door = union{
    box { <-0.05, -1, -0.5>, <0.05, 0.5, 0.5> }
    cylinder { <-0.050001, 0.5, 0>, <0.050001, 0.5, 0>, 0.5 }
}
    
// TOWER
#declare tower = union {

    difference{
        
        // "stuff we add to the basic box of the tower"
        union{
            box { <-3,   0, -3> < 3, 12,  3>  }
            object {door scale < 2, 1.1, 1.2>  translate < 3, 8, 0>  }
            object {door scale < 2, 1.1, 1.2>  translate < 3, 8, 0> rotate -90*y}
            object {door scale < 2, 1.1, 1.2>  translate < 3, 1, 0> }
    
        } // end of basic tower "adds"
    
        // "stuff we "chop off" the basic tower
        box { <-2,  -1, -2> < 2, 13,  2>  }
       
        #declare ang = 0;
        #while (ang < 360)
                #declare along = -2;
                #while (along <= 2)
                    object {wedgie scale < 0.5, 0.4, 0.5> 
                                   translate < along, 11.5, -1.5> 
                                   rotate ang*y}
                    #declare along = along + 1;
                #end  
        #declare ang = ang + 90;
        #end
    
    
        #declare ang = 0;
        #while (ang < 180)
            #declare along = -1;
            #while (along <= 1)
                object {wedgie scale < 0.5, 0.4, 0.5> 
                               translate < along, 8.0, -1.5> 
                               rotate ang*y}
                object {wedgie scale < 0.5, 0.4, 0.5> 
                               translate < along, 5.0, -1.5> 
                               rotate ang*y}
                #declare along = along + 2;
            #end  
        #declare ang = ang + 90;
        #end 
        
        object {door scale < 25, 1, 1>  translate < 3, 8, 0> }
        object {door scale < 25, 1, 1>  translate < 3, 8, 0> rotate -90*y }
        object {door scale < 25, 1, 1>  translate < 3, 1, 0> }
        
    } // end of chop off stuff



    
    //actual door 
    object {door scale < 0.05, 1, 1>  translate < 2.9, 8, 0> pigment {Brown} }
    object {door scale < 0.05, 1, 1>  translate < 2.9, 8, 0>  rotate -90*y pigment {Brown} }
    object {door scale < 0.05, 1, 1>  translate < 2.9, 1, 0> pigment {Brown} }
    
    // top floor
    box { <-2.5, -0.1, -2.5>, <2.5, 0, 2.5> translate 10.5*y pigment {Brown}}

}  //end of TOWER declare object
    
    
object {tower 
   
   pigment {Grey}

}  

The first thing to try is a thing called a “normal”. As 
mentioned earlier, the mathematical term “normal” 
basically means “at right angles to” specifically, a plane 
or surface. In vector mathematics it's used to describe 
the gradient of a surface at a specific point using the 
vector normal to that gradient. That's it for the maths!
In POVRay a “normal” is a function where it generates 
extra shading “as if” there were a changing gradient 
across our mathematically pure flat surfaces.
It “spoofs” a surface that isn't flat. 
Here's a piccy of what we rendered last time.

Our “new” version of the code now relegates all texturing, 
colouring and spoofing of the grey stuff in the image to 
the tower call instruction only. (If you render the image with no 
pigment statement at all in that call statement, it all comes out black, 
apart from the “woody” bits.)
So, let's try a function called “bumps”

object {tower 
   
   normal {bumps 1}
   pigment {Grey}

} 

“bumps” is a built in POV normal function. Here I've 
called it up at a default strength of 1. The strength 
identifies the depth of “shading” used in some internal 
manner. One tends to “play” with these things. 
Unstated here is the scale which defaults to 1.
So if we reduce the scale to 0.1 manually.
object {tower 
   
   normal {bumps 1 scale 0.1}
   pigment {Grey}

} 

Which looks pretty cool already, I think.



Wacking the bumps strength up to something like 3 gives 
it a rather nice Norman conquest flinty look?

object {tower 
   
   normal {bumps 3 scale 0.1}
   pigment {Grey}

} 

But that's not all! There are other around 32 other 
“normal” functions coded into POV, like agate, brick, 
bozo, checker and so on. One particular one of note is 
“crackle”.

object {tower 
   
   normal { crackle 1 scale 0.5 }

   pigment {Grey}

}  

Now, the Help files (which are available as part of the 
software, not some download thing! Aaaaand are really 
really good! ) will talk about Voronoi functions and stuff, 
which is probably cool for not being interesting at parties 
but does make for nice masonry!

Let's zoom in a sec, going to “angle 10” on the camera!
Do you see where I'm going with this?

No? Well, most of the time, neither do I.

This is where POVRay can get really complicated, or, in 
my case, incomprehensible!  

Let's see if I can explain at least some of this. 
POVRay uses “patterns”. The “crackle” used above is a form of pattern that in this case is used to affect the 
shading of the flat surface, but that isn't all the pattern can do. POVray also has things called
color_map, 
pigment_map
texture_map
(and image_map) 
it may have maps I haven't tried yet! 
Anyway. The patterns return values between 0 and 1 for different parts of the pattern which can be exploited for 
colouring/pigmenting, texturing and so on. They can even be “nested”. It's not just complicated – it's FUN.



So let's remove (comment out) our normal command 
here and try this

object {tower 
   
   pigment {crackle 
      scale 0.5
      pigment_map {
          [0.00 color Gray90 ]
          [0.10 color Gray50 ] }
    }        
   
//   normal { crackle 1 scale 0.5 }
//   pigment {Grey}

}  

That's kinda interesting. What exactly happened here?
We asked for a pigment using a pattern “crackle” (incidentally scaled at 0.5 to match our normal earlier)
In that pattern we said 
if the pattern value is 0.00 to 0.10 then make it light grey “Gray90”, above 0.1 make it darker “Gray50”
As such we are discovering something about the pattern “crackle”. If you increase the 0.1 to (say) 0.5 then the 
“mortar” lines spread out further (and incidentally, tend to fade out rather than have a decent “edge”)

(You can have multiple ranges in this statement, you just add more [0.xx color <something>] lines.)
It gets better. 
The line [0.1 color Gray50] has effectively isolated a band 
of the pattern we can do some more with. We can nest 
another map in here.
This time, I'm going to use some colourful colors. 
Perhaps to illustrate a point.
   
   pigment {crackle 
      scale 0.5
      pigment_map {
          [.00 color Gray90 ]
          [.10 crackle 
              solid
              pigment_map { 
                  [0.0 color Red] 
                  [0.2 color Orange] 
                  [0.4 color Yellow] 
                  [0.6 color Green] 
                  [0.8 color Blue]     }    ]
     
       }      

    }        
But mostly because I have trouble understanding 
everything going on here!
Let's see If I get this right.
We've taken one of our […...] pigment map elements (the one equivalent to our “masonry blocks”) and made 
another crackled pigment_map in there!
An important command added to this new map is “solid”, which, as far as I understand, makes the block all one 
pattern value (or color, here). That I can understand. It appears to choose a roughly random seeming value. 
Our pigment_map then colours those blocks. The net result is blocks of uniform colour. 
(What I don't understand is the fact that you get different shades of those colours! Perhaps I need to read the help 
more!) 
Anyway, it works for us. 
If we change the colours and add the normal back in 
again
   pigment {crackle 
      scale 0.5
      pigment_map {
          [.00 color Gray90 ]
          [.10 crackle 
                 solid
                pigment_map { 
                    [0.0 color Gray05] 
                    [0.2 color Gray20] 
                    [0.4 color Gray35] 
                    [0.6 color Gray50] 
                    [0.8 color Gray65] }    ]
        
        }      
    }        
   
   normal { crackle 1 scale 0.5 }



Alternatively, you could use the bumps normal we played 
with earlier together with the crackle pigment map.
object {tower 

   pigment {crackle 
      scale 0.5
      pigment_map {
          [.00 color Gray90 ]
          [.10 crackle 
                 solid
                pigment_map { 
                    [0.0 color Gray05] 
                    [0.2 color Gray20] 
                    [0.4 color Gray35] 
                    [0.6 color Gray50] 
                    [0.8 color Gray65] }    ]
           
                    
        }      
    }        
   
   normal { bumps 1 scale 0.5 }

There are quite a few things we could play with here. 
More than I can cover in this tutorial, but that's part of the 
point really, exploration of the possibilities is a lot of the 
fun with POVRay.

What I will mention here is that  this pattern  “wraps” 
around objects quite well. This is not always the case, the 
“brick” pattern only really works well with one side, so you 
might want to create 1 wall separately, pigment_map 
that, and then rotate “wall and all” together to create 
rooms.

If we wanted to, we could apply a different pigment map to our door 
surrounds (for example). Other pigment_maps are available for wood, 
metal, marble... all sorts.

If that wasn't confusing (and I hope it wasn't) POVRay uses a number of terms/keywords to differentiate colors, 
pigments, textures, and materials (in order of complexity, I think)
Added to that you can also add a “finish” – for example  “phong” which defines how shiny something is. As we've 
seen earlier, we can also define transparency of colours/objects. 
I could probably write a complete new Tutorial just on that subject, but I'd have to pretend that I fully understand it 
all myself! Lot's of the fun I get from POVray is serendipity – the discovery of new things by chance, often.

If I REALLY want to create something specific, then yeah, I go out on the web or read the Help (even I sometimes 
“read the manual”). 

Image_Maps
Another method for “painting” our tower is to use actual 
images. This is a very handy way of doing things but, like 
all 3D programs, there are problems with “wrapping” solid 
objects with a 2D image. POVRay is not much different 
and in general it is better to image_map individual flat 
surfaces separately before  building up a 3D shape with 
them. Sort of.

Anyhow, for this tutorial I'm using this PNG image

Erm. Rather than me having to attach a PNG file to this tutorial in 
some way you could try a screen shot of this or just use some image 
of your own. To keep the tutorial consistent it would help if your image 
was 200 x 200 pixels.

So how?
Well, put the image you want to use in the same directory 
as our POV code. (You can set up separate directories 
for this sort of thing but you would need to add the area 
as a Library in the INI file)

Comment out all our “normal” command stuff and put in 
the image_map 

pigment {image_map  { png "DirtCobble.png"} }
    
/*
   pigment {crackle 
      scale 0.5
      pigment_map {



So what do we have here?

Well, as I understand it, POVRay takes our image and, 
by default, “projects” it on (and sort of through?) the solid 
object from one direction. By default it scales the image 
to fit on a unit cube, and repeats it in all directions at that 
scale. 
So, here the new pattern looks a bit small – compared to 
our “normal” method of doing it.

So by changing the scale

pigment {image_map  { png "DirtCobble.png"} scale 5 }

Which is kinda nice but rather one sided. This isn't 
perfectly solvable the way I've constructed this tower. As 
with the “brick” normal and others it's probably best to 
make the wall alone, texture it, then replicate and rotate 
them to make the tower.

However, a slightly “cheaty” way to get something out of 
this is to rotate the image_map.

pigment {image_map  { png "DirtCobble.png"}     
    scale 5
    rotate 45*x
    rotate 45*y
    }

Here, I've rotated the image_map to “project” our png 
image from 45 degrees, thus keeping the image from 
“smearing” along the edges parallel to the z axis.
As you can see though, this is at the expense of partly 
smearing the image on all sides.

There is a technique for wrapping an image_map around an object using the “once” command 
e.g.
pigment {image_map  { png "some_image.png" once } }

This is declared inside the weird brackets along with the image filename and is best used on cylinders and the like. 
I'm not going to go into this here, you can experiment yourself, but you need an image of suitable size and 
resolution for the object as the image, by definition, doesn't get repeated.

Anyway, that's my little introduction to “colouring things” for POVRay.


